NITTE EDUCATION TRUST | NMAM INSTITUTE OF TECHNOLOGY

Course : Microprocessor and Microcotroller Lab.

A Mini Project Report on

ULTRASONIC RANGEFINDER USING 8051

Submitted By,

NAME	USN
Aneesh Acharya	4NM18EC021
Anirudh Nayak	4NM18EC022
Ankitha A G	4NM18EC023
Ankitha P Shetty	4NM18EC024

Under the guidance of ,

Mr.Sukesh Rao M

DEPARTMENT OF ELECTRONICS & COMMUNICATION ENGINEERING

N.M.A.M. INSTITUTE OF TECHNOLOGY, NITTE - 574110

2020 – 2021

ABSTARCT

Ultrasonic rangefinder measures distance by emitting a pulse of ultrasonic sound that travels through the air until it hits an object. When that pulse of sound hits an object, it's reflected off the object and travels back to the ultrasonic rangefinder. Ultrasonic rangefinder can measure distance based on time. AT89S52 microcontroller and the ultrasonic sensor module HC-SR04 forms the basis of this circuit. The ultrasonic module sends a signal to the object, then picks up its echo and outputs a wave forms whose time period is proportional to the distance. The microcontroller accepts this signal, performs necessary processing and displays the corresponding distance on the LCD display. Precision of this setup is one centimetre.

II. OBJECTIVES OF THE PROJECT

a. To trigger the ultrasonic sensor using 8051.

b. Calculate distance using internal Timers of 8051.

c. Display result on the LCD display.

III.BLOCK DIAGRAM



The ultrasonic sensor has 2 pins (Trig,Echo) connected to the 8051. The Trig Pin is used to activate the sensor on demand by 8051, the Echo is the output of the sensor which has distance related information.

LCD has a total of 11 pins connected to 8051, out of which 3 pins are for Control and the rest 8 are data lines. The control pins are Register Select (RS), Read/Write (RW) and Enable (E). Data lines are D0- D7.

Actual connnections in the project:

Trig (sensor) - P 3.0

Echo (sensor) - P 3.1

RS (LCD) - P 3.5

RW (LCD) - P 3.6

E (LCD) - P 3.7

D0-D7 (LCD) - PORT 2

IV. Method

Circuit Diagram:



a. To trigger the ultrasonic sensor using 8051.



The ultrasonic sensor needs a trigger pulse of minimum 10 µS to transmit the sound waves and to produce ECHO pulse. For the distance to be updated continous stream of pulses are fed to trigger pin with required amount of delay. This process is repeated indefintely in the project. The ECHO from the sensor is used to start and stop Internal TIMER 1 of 8051.

From the timing diagram it is noted that the cycle time for the sensor is 50 µS. Therfore any trigger pulse applied within 50 µS preceding a trigger will not be accounted by the sensor. It takes 58 µS for sound waves to return to sensor after reflecting from a surface 1 cm away. Assuming speed of sound to be 340 m/S time to transit 1cm will be,

$$time = \frac{2 * distance}{speed}$$

$$time = \frac{2 * 1}{34000} = 58.8\,\mu S$$

The distance is conidered twice since sound waves are detected after reflection from the surface.This is total transit time , to and fro from the surface. The pulse width of the ECHO signal is the transit time.

b. Calculate distance using internal Timers of 8051.

Flow chart:



The TIMER1 is started when ECHO pulses goes high and timer runs until the ECHO pulse is available. The TIMER 1 is used in 8 bit auto-reload mode. The reload value is 207, which means after overflow the TL 1 is reloaded with 207. Hence the timer only counts 255-207=48 counts before overflow. After every overflow accumulator is incremented, the value in the accumulator is the distance in centimetre. It is seen from earlier calculation that 58 µS translates to one centimetre, here a value of 48 only is considerd because there are instrcutions between start and stop of timer which account for the 10 µS delay. So the code scans through the ECHO pulse in steps of 48 counts and after each step it increments the value in accumulator. Everytime the timer overflows it starts again from 207 count.The flow chart given above accurately describes the process. So once the ECHO pulse ends the value in the accumulator is the distance in centimetres. This value is temporarily stored in a different register , later to be printed on the LCD.

c. Display result on the LCD display.

To display output on LCD, it's initialized as a screen with two rows and 5X7 (space to fully display a single letter or number) matrix by sending 38H as a command to it via 8051 microcontroller.



Then the screen is cleared of previously stored values by sending 01H as another command. Now first address location of data to be printed on LCD is sent to it via a command i.e. 80H (points the cursor to first location in first row). In case we need to send data to first location in second row, the command to be sent is 0C0H.Then the outputs to be displayed are sent as Data, be it a string or and any other data type(manual conversion is required). Every time a new element is printed on the screen, the cursor moves to next location to print next element in the data.

If a string is to be displayed on the screen then there is no need for the designer to convert those data to ASCII (format in which LCD interprets data) manually. In case we want to display a data which is not static like a string then we need to convert it to its equivalent ASCII format. In this project, each digit of the distance to be displayed is unpacked as separate digits and converted to its ASCII format via 8051 using the table below and sent to LCD as Data, as and when we receive inputs from the sensor.

Key(unpacked)	ASCII(hex)
0	30
1	31
2	32
3	33
4	34
5	35
6	36
7	37
8	38
9	39

V. Result

Experimental setup :



Test conditions:

1. Target is kept 3cm from the sensor, control to the experiment is a standard 30 cm measuring scale. The sensor can't measure below 3cms because of its construction. The distance value is stored in an 8 bit register so the maximum value that can theoretically be measured with the sensor is 255 cm.



2. Target kept at 12 cm from the sensor,



Code Snippet:

```
//Ultrasonic Rangefinder.
ORG 00H
MOV P0, #0H                 //initializing p0 & p1 as output ports
MOV P2, #0H
MOV A, #38H                 //initializing lcd display as 2 rows and
5x7 matrix.
ACALL CMD                   //calling subroutine 'CMD'
MOV A, #01H                 //clear LCD display
ACALL CMD
MOV A, #80H                 //move cursor to address '80H'
ACALL CMD
MOV DPTR,#FIRST     //TO display the string 'ultrasonic'
FIR: MOV A,#00H
    MOVC A,@A+DPTR
JZ OP
ACALL DAT
INC DPTR
SJMP FIR
OP:
MOV A, #0C0H            //TO display the string 'rangefinder'
ACALL CMD
MOV DPTR, #SECOND
SEC: MOV A,#00H
    MOVC A,@A+DPTR
JZ OP1
ACALL DAT
INC DPTR
SJMP SEC
OP1:
MOV R7,#10                  // call delay of 1.42 sec
RPT:LCALL DELAY2
DJNZ R7,RPT
 MOV A,#01H
ACALL CMD
MOV A,#80H
ACALL CMD
MOV DPTR,#DIST       //to display the string 'Distance:'
DIS: MOV A,#00H
    MOVC A,@A+DPTR
JZ OP2
ACALL DAT
INC DPTR
SJMP DIS
OP2:
MOV A,#8DH                  //move cursor to display cms(centimeters)
ACALL CMD
MOV DPTR,#CMS
CM: MOV A,#00H
    MOVC A,@A+DPTR
JZ OP4
ACALL DAT
INC DPTR
SJMP CM
// code for measuring distance using HCSR04 (ultrasonic sensor)
OP4:
CLR P3.0
SETB P3.1                       //initialize pin p3.1 as input port
(echo pin)
MOV TMOD,#00100000B     //to initialize Timer1 in 8 bit auto
reload mode
MAIN: MOV TL1,#207D     // store initial count value in TL1
     MOV TH1,#207D          //stores count reload value
     MOV A,#00000000B
     SETB P3.0                  //sends a trigger of width 10us
     ACALL DELAY1
     CLR P3.0
HERE: JNB P3.1,HERE         //wait till echo becomes High
BACK: SETB TR1              //start  Timer1
HERE1: JNB TF1,HERE1    //wait till overflow
     CLR TR1                //stop timer
     CLR TF1                //clear overflow flag
     INC A                  //increments accumulator for every
overflow
OUT:  JB P3.1,BACK          //loops until echo becomes low
     MOV R4,A               //stores distance
     ACALL DLOOP
     SJMP MAIN
DLOOP: CLR P1.3             //subroutine for displaying value in
LCD
      CJNE R4,#16,GO
      GO: JC GO1
   SETB P1.3
GO1:   MOV A,#89H
      ACALL CMD
      MOV DPTR,#ASCII       //lookup table for ASCII numbers
      MOV A,R4              //to unpack the distance
   MOV B,#100D
   DIV AB
   MOVC A,@A+DPTR
   ACALL DAT
   MOV A,B
   MOV B,#10D
   DIV AB
   MOVC A,@A+DPTR
   ACALL DAT
   MOV A,B
   MOVC A,@A+DPTR
   ACALL DAT
   MOV R5,#0FFH
      L2:ACALL DELAY        //Delay to make output data legible
   DJNZ R5,L2
   RET
 CMD: CLR P3.5              //subroutine to send commands to LCD
    CLR P3.6
MOV P2,A
SETB P3.7
ACALL DELAY
CLR P3.7
RET
DAT: SETB P3.5              //subroutine to send commands to
LCD
    CLR P3.6
MOV P2,A
SETB P3.7
ACALL DELAY
CLR P3.7
RET
DELAY1: MOV R6,#2D
LABEL1: DJNZ R6,LABEL1
       RET
DELAY : MOV R1,#0FFH
     L1:  DJNZ R1,L1
   RET
DELAY2: MOV R1,#0FFH
       L12: MOV R3,#0FFH
L21: DJNZ R3,L21
DJNZ R1,L12
RET
DIST: DB "Distance:",0
FIRST: DB "Ultrasonic",0
SECOND : DB "Rangefinder",0
CMS: DB "cms",0
ASCII : DB 30H,31H,32H,33H,34H,35H,36H,37H,38H,39H
END
```